



# DECUS

## PROGRAM LIBRARY

DECUS NO.	8-263
TITLE	XYPLOT - A VERSATILE PLOT ROUTINE FOR THE D/A CONVERTER
AUTHOR	Eugene E. Wells, Jr.
COMPANY	U.S. Army Electronics Command Fort Monmouth, New Jersey
DATE	March 27, 1970
SOURCE LANGUAGE	PAL-D

### ATTENTION

This is a USER program. Other than requiring that it conform to submittal and review standards, no quality control has been imposed upon this program by DECUS.

The DECUS Program Library is a clearing house only; it does not generate or test programs. No warranty, express or implied, is made by the contributor, Digital Equipment Computer Users Society or Digital Equipment Corporation as to the accuracy or functioning of the program or related material, and no responsibility is assumed by these parties in connection therewith.

DECUS

1953-54



Name	Address
John Doe	123 Main St
Jane Smith	456 Elm St
Bob Johnson	789 Oak St
Alice Brown	101 Pine St
Charlie White	202 Cedar St
Diana Green	303 Birch St
Frank Black	404 Spruce St
Grace Hall	505 Willow St
Henry King	606 Ash St

THE UNIVERSITY OF CHICAGO  
LIBRARY  
1200 SOUTH EAST ASIAN BLVD  
CHICAGO, ILL. 60607



# XYPLOT - A VERSATILE PLOT ROUTINE FOR THE D/A CONVERTER

DECUS Program Library Write-up

DECUS No. 8-263

## ABSTRACT

XYPLOT is a one page subroutine which scales, offsets, and simultaneously plots two integer arrays from any two data fields, using the D/A converter and a conventional x-y recorder. Incorporation of a programmed pen lift makes available to those users having the necessary additional hardware a point plot capability, selectable by properly setting bit zero of the variable PENFLG in the calling sequence. Loading of the additional 41 (octal) memory location program GENX, and setting of bit one of PENFLG allows a y-data array to be plotted versus locally generated, equispaced x-data, with the spacing interval user specified.

The program was primarily designed to conveniently handle the output of real experimental data with the minimum of prior manipulation consistent with generality. Scaling of the data, for example, from internal code integers to meaningful units is accomplished by XYPLOT just prior to the actual plot step, a procedure which obviates an intermediate, storage consuming, array scaling step. Flexibility of plot size, on the other hand, is provided by passing offset and axis scale values via the subroutine argument list. Thus, these values may be user supplied, or may be computed by the calling program just prior to the call of XYPLOT.

Minimum Hardware Required: PDP-8 with AA-01A D/A Converter (Mod AH03)

Other Programs or Subroutines Needed: Floating Point Package III (DEC-08-YQ3A-PB), Subroutines FETCH, FIX, and FLOAT (supplied).

Storage Requirement: 1 page, plus 41 (octal) locations if x-axis data is to be generated locally by the subroutine at plot time.

Execution Time: Limited by the number of wait loops used. Adequate delay is provided for the x-y recorder to settle between points.

```

*XYPLOT
/SUBROUTINE XYPLOT
/
/CALLING SEQUENCE:
/      100      EFFECTIVE JMS XYPLOT
/      101      "ADDRESS OF X OFFSET (XOFF)
/      102      "ADDRESS OF Y OFFSET (YOFF)
/      103      "ADDRESS OF XAXIS UNITS/IN (XPLTSC)
/      104      "ADDRESS OF YAXIS UNITS/IN (YPLTSC)
/      105      "ADDRESS OF INTEGER>UNITS FACTOR (XFACTR)
/      106      "ADDRESS OF INTEGER>UNITS FACTOR (YFACTR)
/      107      "FLAG: BIT ZERO SET=PLOT POINTS
/              BIT ONE SET=GENERATE X-DATA
/      110      "ADDRESS OF NO DATA POINTS
/      111      "ADDRESS OF DATA FIELD FOR XDATA
/      112      "BASE ADDRESS OF XDATA, OR POINT SPACING
/      113      "ADDRESS OF DATA FIELD FOR YDATA
/      114      "BASE ADDRESS OF YDATA
/
/THE SUBROUTINE SCALES DATA TO ANY DESIRED SIZE AND
/LOTS EITHER POINT BY POINT OR WITH THE PEN
/CONTINUOUSLY DOWN. THE PEN LIFTS AFTER THE PLOT
/IS COMPLETE AND RETURNS TO ITS ZERO POINT.
/
/REQUIRED ANCILLARY PROGRAMS:
/      FLOATING POINT PACKAGE
/      FIX
/      FLOAT
/      FETCH
/
/PAGE ZERO LOCATIONS:
/      AUTOINDEX 12,13
/
/THE SCALING ALGORITHM:
/
/       $INTEGER\ TO\ DAC = (XDATA * XFACTR + XOFF) / XPLTSC * DACSCL$ 
/
/       $DACSCL = VOLTS\ OUT / (INTEGER\ IN * RECORDER\ SCL\ (V/IN$ 
/
/THE DATA IS ASSUMED TO BE INTEGER, THE SCALING
/CONSTANTS FLOATING. IF LOCAL GENERATION OF X DATA
/IS DESIRED, LOAD GENX, SET BIT ONE OF PENFLG, AND
/NOTE THAT XDTABS (IN THE ARGUMENT LIST) NOW
/MUST CONTAIN THE ADDRESS (IN THE SAME DF AS THE
/IF) OF THE FLOATING MODE POINT SPACING.
/
/PNLIFT IS A CONSTANT WHICH MAY BE PREADJUSTED TO
/PROVIDE PROPER VOLTAGE FOR DRIVING A PEN LIFT
/RELAY.
/

```



0200	0000	XYPLOT, 0	
0201	4465	JMS I FETCHP	
0202	0014	14	
0203	0000	XOFF, 0	
0204	0000	YOFF, 0	
0205	0000	XPLTSC, 0	
0206	0000	YPLTSC, 0	
0207	0000	XFACTR, 0	
0210	0000	YFACTR, 0	
0211	0000	PENFLG, 0	
0212	0000	POINTS, 0	
0213	0000	DFXDTA, 0	
0214	0000	XDTABS, 0	
0215	0000	DFYDTA, 0	
0216	0000	YDTABS, 0	
0217	1211	TAD PENFLG	
0220	0377	AND (2000	
0221	7640	SZA CLA	/GEN XDATA?
0222	5226	JMP OMIT	/YES.
0223	1613	TAD I DFXDTA	/NO. SET UP X POINTER
0224	1360	TAD K6201	
0225	3253	DCA DFX	
0226	1615	OMIT, TAD I DFYDTA	
0227	1360	TAD K6201	
0230	3267	DCA GETY	
0231	6214	RDF	
0232	1360	TAD K6201	
0233	3255	DCA RSTRE	
0234	1255	TAD RSTRE	
0235	3271	DCA REST	
0236	1612	TAD I POINTS	
0237	7041	CIA	
0240	3354	DCA CNTR	
0241	7240	CLA CMA	
0242	1614	TAD XDTABS	
0243	3012	DCA 12	
0244	7240	CLA CMA	
0245	1616	TAD I YDTABS	
0246	3013	DCA 13	
0247	1211	RPEAT, TAD PENFLG	
0250	0377	AND (2000	
0251	7640	SZA CLA	/GENERATE XDATA?
0252	5765	JMP I GENXP	
0253	7402	DFX, 7402	
0254	1412	TAD I 12	/NO. FETCH XPOINT
0255	7402	RSTRE, 7402	
0256	4463	FLOAT	
0257	4407	JMS I 7	/ENTER FLOATING INT
0260	3607	FMPY I XFACTR	
0261	1603	FADD I XOFF	
0262	4605	FDIV I XPLTSC	

0263	4355		FDIV DACSCL	
0264	0000		FEXT	
0265	4464		FIXPT	
0266	3361		DCA XTEMP	
0267	7402	GETY,	7402	
0270	1413		TAD I 13	/FETCH Y POINT
0271	7402	REST,	7402	
0272	4463		FLOAT	
0273	4407		JMS I 7	
0274	3610		FMPY I YFACTR	
0275	1604		FADD I YOFF	
0276	4606		FDIV I YPLTSC	
0277	4355		FDIV DACSCL	
0300	0000		FEXT	
0301	4464		FIXPT	
0302	3362		DCA YTEMP	
0303	4342		JMS PLOT	/POSITION PEN
0304	4336		JMS WAIT	
0305	7000		NOP	
0306	1363		TAD PNLIFT	
0307	6554		DAL3	/PEN DOWN.
0310	7300		CLA CLL	
0311	4336		JMS WAIT	/TO BE SURE PEN IS DOWN
0312	7000		NOP	
0313	1211		TAD PENFLG	
0314	7710		SPA CLA	/POINT PLOT?
0315	6554		DAL3	/YES. PEN UP
0316	2354		ISZ CNTR	/NO. LAST POINT?
0317	5247		JMP RPEAT	/NO.
0320	6554		DAL3	/YES. PEN UP
0321	4336		JMS WAIT	
0322	1361	RTNZRO,	TAD XTEMP	/RETURN PEN TO ZERO.
0323	7110		CLL RAR	/DIVIDE BY TWO
0324	3361		DCA XTEMP	
0325	1362		TAD YTEMP	
0326	7110		CLL RAR	/DIVIDE BY TWO
0327	3362		DCA YTEMP	
0330	4342		JMS PLOT	
0331	1361		TAD XTEMP	
0332	1362		TAD YTEMP	
0333	7640		SZA CLA	/BOTH AXES ZERO?
0334	5322		JMP RTNZRO	/NO. DO AGAIN
0335	5600		JMP I XYPLOT	/YES. RETURN
0336	0000	WAIT,	0	
0337	2364		ISZ WAIT1	
0340	5337		JMP .-1	
0341	5736		JMP I WAIT	
0342	0000	PLOT,	0	
0343	4336		JMS WAIT	
0344	1361		TAD XTEMP	
0345	6551		DAL1	/LOAD DAC CH(1) WITH X



0346	7200	CLA	
0347	1362	TAD YTEMP	
0350	6552	DAL2	/LOAD DAC CH(2) WITH Y
0351	7200	CLA	
0352	4336	JMS WAIT	/SETTLE TIME
0353	5742	JMP I PLOT	
0354	0000	CNTR, 0	
0355	7771	DACSCL, 7771;	
0356	2400	2400;	
0357	5002	5002	/10.0V/2047*1V/IN (DECIMAL)
0360	6201	K6201, 6201	
0361	0000	XTEMP, 0	
0362	0000	YTEMP, 0	
0363	0000	PNLIFT, 0	/ADJUST TO OPERATE PEN RELAY
0364	0000	WAIT1, 0	
0365	0400	GENXP, GENX	
0377	2000		
		*PAGE	/TRICKS ASSEMBLER TO PRINT LITERALS AND LINE
		/	
		/NEED BE LOADED ONLY TO GENERATE LOCAL XDATA	
		/	
		*GENX	
0400	1632	GENX, TAD I FLGP	/USE AS FLAG: NON-ZRO FIRST PASS
0401	7650	SNA CLA	/FIRST PASS?
0402	5216	JMP INCR	/NO. INCREMENT
0403	3633	DCA I XTEMP	/YES. INITIALIZE
0404	3632	DCA I FLGP	/CLEAR FLAG
0405	3237	DCA XVAR	
0406	3240	DCA XVAR+1	
0407	3241	DCA XVAR+2	
0410	7001	IAC	
0411	1012	TAD 12	/POINT SPACING ADDRESS
0412	3230	DCA PSPACE	
0413	1634	TAD I XPLTSP	
0414	3231	DCA XSCALE	
0415	5636	JMP I GETYP	
0416	4407	INCR, JMS I 7	
0417	5237	FGET XVAR	
0420	1630	FADD I PSPACE	
0421	6237	FPUT XVAR	
0422	4631	FDIV I XSCALE	/FETCHES XPLTSC
0423	4635	FDIV I DACSCP	
0424	0000	FEXT	
0425	4464	FIXPT	
0426	3633	DCA I XTEMP	
0427	5636	JMP I GETYP	
0430	0000	PSPACE, 0	
0431	0000	XSCALE, 0	
0432	0215	FLGP, DFYDTA	
0433	0361	XTEMP, XTEMP	/POINTER TO XTEMP
0434	0205	XPLTSP, XPLTSC	/POINTER TO XPLTSC
0435	0355	DACSCP, DACSCL	/POINTER TO DACSCL
0436	0267	GETYP, GETY	/POINTER TO RETURN
0437	0000	XVAR, 0;	
0440	0000	0;	
0441	0000	0	

CNTR 0354  
DACSCCL 0355  
DACSCP 0435  
DFX 0253  
DFXDTA 0213  
DFYDTA 0215  
FETCHP 0065  
FIX 4464  
FLGP 0432  
GENX 0400  
GENXP 0365  
GETY 0267  
GETYP 0436  
INCR 0416  
K6201 0360  
OMIT 0226  
PENFLG 0211  
PLOT 0342  
PNLIFT 0363  
POINTS 0212  
PSPACE 0430  
REST 0271  
RPEAT 0247  
RSTRE 0255  
RTNZRO 0322  
WAIT 0336  
WAIT1 0364  
XDTABS 0214  
XFACTR 0207  
XOFF 0203  
XPLTSC 0205  
XPLTSP 0434  
XSCALE 0431  
XTEMP 0361  
XTEMPPP 0433  
XVAR 0437  
XYPLOT 0200  
YDTABS 0216  
YFACTR 0210  
YOFF 0204  
YPLTSC 0206  
YTEMP 0362



/DEFINITIONS FOR ASSEMBLY OF XYPLOT.

/

XYPLOT=200

GENX=400

FIX=JMS I 64

FLOAT=JMS I 63

DAL1=6551

DAL2=6552

DAL3=6554

FETCHP=65

/

PAUSE

```

*FETCH
/SUBROUTINE FETCH
/
/FETCHES ARGUMENTS FOR SUBROUTINES
/
/CALLING SEQUENCE:
/      200      SUBR,      0
/      201      EFFECTIVE JMS FETCH
/      202      NUMBER OF ARGUMENTS (OCTAL)
/      203      ARG1,      0
/      204      ARG2,      0
/      .
/      .      ETC., ONE FOR EACH ARGUMENT
/      .      NEXT EXECUTABLE STATEMENT
/      20-
/
/FETCH UPDATES THE EXIT OF THE CALLING SUBROUTINE
/PAGE ZERO LOCATIONS REQD:
/      AUTOINDEX 16, 17
/      65
/
/
/
FETCH, 0
      CLA CMA
      TAD FETCH      /POINTER TO STORE ARGUMENTS
      DCA 16
      CLA CLL CMA RAL /SET ACC TO MINUS TWO
      TAD FETCH      /POINTER TO SOURCE ARG LIST
      DCA FETCH
      CMA
      TAD I FETCH    /GET ARG SOURCE LIST ADDRESS
      DCA 17
      TAD I 16        /FETCH NO OF ARGUMENTS
      CIA
      DCA CNTR
      NEXT, TAD I 17   /FETCH THE ARGUMENT
      DCA I 16        /STORE IT
      ISZ CNTR
      JMP NEXT
      CLA IAC
      TAD 17
      DCA I FETCH    /UPDATE SUBROUTINE EXIT
      JMP I 16       /RETURN TO SUBROUTINE
CNTR, 0
*65
      FETCH
$

```



\*FIX  
/SUBROUTINES TO TRANSMIT DATA FROM FAC TO ACC AND BACK  
/FIX OR FLOAT THE RESULTING DATA.  
/

/FOR USE WITH FLOATING POINT PACKAGE III  
/

FIX,       0  
          JMS I FIXPTR  
          TAD 45  
          JMP I FIX

FIXPTR, 4757

FLOA,       0  
          DCA 45  
          JMS I FLOPTR  
          JMP I FLOA

FLOPTR, 5563

\*63

FLOA  
FIX

PAUSE

